

\$online

/\*

*GAMS program used to estimate capacity with constant returns to scale. These models are based on the Johansen (1968) concept of capacity, which can be found in the text "Production Frontiers", by Fare, Grosskopf and Kokkelenberg (1994), page 261-263.*

Author: John B. Walden

NMFS/NEFSC

166 Water St.

Woods Hole, MA 02543

(508) 495-2355

John.Walden@Noaa.Gov

*This version will estimate capacity for each vessel, based on an output oriented DEA model.*

\*/

*/\* The following line turns off listing of some elements in the GAMS listing file\*/*

\$OFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF

*/\*NEXT DEFINE INPUTS AND OUTPUTS\*/*

SET INOUT /out1, out2, out3, out4,out5, out6, fix1, fix2, fix3, var1, var2/

OUTPUT(INOUT) /out1\*out6/

FIXED(INOUT) /fix1, fix2, fix3/

VAR(INOUT) /var1, var2/

OBS /1\*200/

SUBOBS(OBS) /1\*82/

ACTOBS(OBS)

*/\*We have allocated enough memory for 200 observations, but our data set only contains 82 observations (subobs)\*/*

*/\*Next, define an alias for the set SUBOBS \*/*

alias (subobs, subobs1)

*/\* The include statement below reads in an external data file which contains a table of observations, inputs and outputs.*

*The offlisting command means that the data won't be included in the listing file.*

\*/

\$OFFLISTING

TABLE ACT(OBS,INOUT) INPUT OUTPUT TABLE

```
$ondelim
$INCLUDE "cap1.csv"
$offdelim
```

```
$ONLISTING
```

```
VARIABLES
```

```
theta          efficiency score
weight(obs)    weights
lambda(obs, var) variable input utilization rate;
```

```
POSITIVE Variable weight, lambda;
```

```
/* The capacity model is defined using three equations. The difference between
this model and the TE output model is that the inputs are divided into
two constraints, one for fixed inputs and one for variable inputs. The variable
Lambda, in the third equation will ensure that the variable inputs
do not constrain production. */
```

```
EQUATIONS
```

```
CONSTR1(OUTPUT, OBS) DEA constraint for each output
CONSTR2(FIXED, OBS) DEA constraint for each fixed input
CONSTR3(VAR, OBS) DEA constraint for variable inputs;
```

```
CONSTR1(OUTPUT, ACTOBS).. SUM(SUBOBS, WEIGHT(SUBOBS)*ACT(SUBOBS,OUTPUT)) =G=
    THETA*ACT(ACTOBS, OUTPUT);
```

```
CONSTR2(FIXED, ACTOBS).. SUM(SUBOBS, WEIGHT(SUBOBS)*ACT(SUBOBS,FIXED)) =L=
    ACT(ACTOBS, FIXED);
```

```
CONSTR3(VAR, ACTOBS).. SUM(SUBOBS, WEIGHT(SUBOBS)*ACT(SUBOBS,VAR)) =E=
    LAMBDA(ACTOBS, VAR)*ACT(ACTOBS, VAR);
```

```
/*Define a parameter to hold results for each pass through
the loop*/
```

```
PARAMETER
```

```
score1(obs)    capacity scores
score2(obs, VAR) optimal variable input utilization rates
;
```

```
/*Define an external file to hold results which tell whether model solved
at each iteration*/
```

```
file caps /cap_res.txt/;
```

```
/*The model defined below consists of three equations.
CONSTR1, CONSTR2, CONSTR3 */
```

```
MODEL CAP /ALL/;
```

```
cap.solprint=2;    /*Turn off writing results to solution file*/
```

```

cap.solveLink=2;    /*Keep model in memory. Improves solution time*/

LOOP(SUBOBS1,

    ACTOBS(OBS)=NO;
    ACTOBS(SUBOBS1)=YES;

    SOLVE CAP maximizing THETA USING LP;

    score1(SUBOBS1) = theta.l;
    score2(subobs1,var)=lambda.l(subobs1,var);

    put caps;

    if ((cap.modelstat eq 1 and cap.solvestat eq 1),

        put @1, subobs1.tl, @10, "optimal", @20, "normal completion" /

    else

        put @1, subobs1.tl, @10, cap.modelstat:>2:0,
            @20, cap.solvestat:>2:0/
    );

);

/*The next file is to output results to a file to be imported
into a spreadsheet program. Results could also be printed to
the listing file with the use of the display command*/

file res /cap_crs.csv/ ;

res.pc=5;
res.pw=50;

put res;

put 'OBS', put 'Theta', loop(var, put var.tl);
put /
loop (subobs1,

    put subobs1.tl, score1(subobs1), loop(var, put score2(subobs1,var)); put /
);
putclose;

```